

大连理工大学本科设计报告

题目： 数字钟

课程名称： 单片机原理及实验 A

学院（系）： 控制科学与工程学部

专 业： 自动化

班 级： 电自 2101

学 号： 20211071423

学生姓名： 李林轩

成 绩：

2024 年 5 月 12 日

题目：数字钟

1. 设计要求

- (1) 显示时间
- (2) 可调整时间


2. 设计分析及系统方案设计

- (1) 初始化及主程序：定时器 T0 工作方式，定时初值，开中断，启动定时。
- (2) 中断服务程序：设置定时，修改计数。
- (3) 向显示缓冲区存放数字：将时以及分送入显示缓冲区。
- (4) 显示子程序：点亮数码管。
- (5) 按键子程序：按键用于调整时间。

3. 主程序中主要变量定义

变量名称	RAM 单元/寄存器	功能
LH	78H	存放时十位
HH	79H	存放时个位
LM	77H	存放分十位
HM	76H	存放分个位
SSet	6FH	用于修改秒计数单元
MSet	70H	用于修改分计数单元
HSet	71H	用于修改时计数单元

4. 主程序程序流程图



5. 程序清单

```

BUF    EQU 6EH
SSet   EQU 6FH
MSet   EQU 70H
HSet   EQU 71H
HM     EQU 76H
LM     EQU 77H
LH     EQU 78H
HH     EQU 79H

```

```

B05    BIT 00H
SET2   BIT P1.1
SET1   BIT P1.0

```

```

ORG    0000H
LJMP   START
ORG    000BH
LJMP   s005
ORG    0100H

```

```

START:    MOV     IE, #00H
          MOV     SP, #30H      ;堆栈指针
          MOV     20H, #00H    ;清标志
          CLR     A
          MOV     BUF, A
          MOV     SSet, A
          MOV     MSet, A

          MOV     HSet, #00H
          MOV     MSet, #01H

          MOV     TMOD, #01H   ;定时器 T0 工作在方式 1
          MOV     TH0, #3CH    ;T0 用于时钟定时, 0.05s 定时初值
          MOV     TLO, #0B0H
          SETB    PTO          ;T0 高级优先
          ACALL   FZ           ;调用向显示缓冲区放数子程序
          ACALL   SHOW        ;调用显示子程序
          LCALL   YS20MS
          SETB    EA          ;开中断
          SETB    ETO
          SETB    TR0         ;启动定时

LOOP1:    ACALL   FZ           ;调用向显示缓冲区放数子程序
          ACALL   SHOW        ;调用显示子程序
          ACALL   KEY         ;调用键盘扫描处理子程序
          SJMP    LOOP1

;=====
;T0 中断服务程序
;=====
s005:     MOV     TH0, #3CH
          MOV     TLO, #0B7H
          PUSH    PSW
          PUSH    ACC
          INC     BUF         ;BUF+1
          MOV     A, BUF
          CJNE    A, #10, s0051
          SETB    B05

s0051:    CJNE    A, #20, DK13 ;累计 20 次为 1s
          CLR     B05         ;清除 0.5s 标志

```

```

MOV BUF, #00H
MOV A, SSet      ;修改秒
INC A
MOV SSet, A
CJNE  A, #60, DK13    ;=60?
MOV SSet, #00H ;等于 60, 计数单元清 0
MOV A, MSet      ;修改分计数单元
INC  A
MOV  MSet, A
CJNE  A, #60, DK13
MOV MSet, #00H
MOV A, HSet      ;修改时计数单元
INC A
MOV HSet, A
CJNE  A, #24, DK13
MOV HSet, #00H
SJMP  DK13
DK13: POP ACC
      POP PSW
      RETI
;=====
;向显示缓冲区放数字程序
;=====
FZ: PUSH  PSW      ;保护现场, (RS1, RS0)
    PUSH  B
    PUSH  ACC

FZ_1: MOV  A, MSet      ;MSet 中内容送入 A
      ACALL CHU10      ;调用除 10 程序
      MOV  HH, B
      MOV  LH, A
FZ_2: MOV  A, HSet      ;HSet 中内容送入 A
      ACALL CHU10      ;调用除 10 程序
      MOV  HM, A      ;分十位送显缓 79H
FZ_3: MOV  LM, B      ;分个位送显缓 78H
FZ_4: POP  ACC
      POP B      ;恢复现场
      POP PSW    ;恢复(RS1, RS0)
      RET      ;子程序返回
;=====
;除 10 子程序
;=====
CHU10: MOV  B, #0AH
        DIV  AB

```

```

    RET
;=====
;显示子程序
;=====
SHOW:  PUSH    PSW
        SETB    RS0        ;转工作寄存器组 1
        CLR     RS1
        PUSH    ACC        ;保护现场
        MOV     A, P2
        ORL     A, #0F0H
        MOV     P2, A      ;关闭显示
        MOV     R0, #HM
        MOV     DPTR, #TABD
        MOV     A, @R0
        MOVC    A, @A+DPTR
        MOV     P0, A
        CLR     P2.4        ;点亮 1 位
        LCALL   YS20MS

        INC     R0
        MOV     DPTR, #TABD
        MOV     A, @R0
        MOVC    A, @A+DPTR
        JNB     B05, DISP_1
        ANL     A, #7FH      ;点亮小数点
DISP_1:  MOV     P0, A
        SETB    P2.4
        CLR     P2.5        ;点亮 2 位
        LCALL   YS20MS

        INC     R0
        MOV     DPTR, #TABD
        MOV     A, @R0
        MOVC    A, @A+DPTR
        MOV     P0, A
        SETB    P2.5
        CLR     P2.6        ;点亮 3 位
        LCALL   YS20MS

        INC     R0
        MOV     DPTR, #TABD
        MOV     A, @R0
        MOVC    A, @A+DPTR
        MOV     P0, A

```

```

    SETB    P2.6
    CLR     P2.7          ;点亮 4 位
    LCALL   YS20MS
    SETB    P2.7
    POP     ACC           ;恢复现场
    POP     PSW           ;返回工作寄存器组
    RET                      ;子程序返回
TABD:  DB   0C0H, 0F9H, 0A4H, 0B0H, 99H, 92H, 82H, 0F8H, 80H, 90H, 0FFH
;=====
;按键子程序
;=====
KEY:   JB   SET1, KEY_3    ;S1 未按下转查 S2
       ACALL DELY10        ;延时消抖
       JB   SET1, KEY_3

KEY_1:  ACALL FZ
       ACALL SHOW
       JNB   SET1, KEY_1    ;等待 S1 键释放
       MOV   A, HSet
       ADD   A, #1
       CJNE  A, #24, KEY_2   ;=24?
       CLR   A

KEY_2:  MOV   HSet, A        ;保存结果
       RET

KEY_3:  JB   SET2, KEY_6    ;S2 未按下转返回
       ACALL DELY10
       JB   SET2, KEY_6

KEY_4:  ACALL FZ
       ACALL SHOW
       JNB   SET2, KEY_4
       MOV   A, MSet
       ADD   A, #1
       CJNE  A, #60, KEY_5
       CLR   A

KEY_5:  MOV   MSet, A

KEY_6:  RET
;=====
;延时子程序
;=====

```

```
DELY10: MOV    R6, #20
```

```
DEL2:  MOV     R7, #7DH
```

```
DEL1:  DJNZ    R7, DEL1  
      DJNZ    R6, DEL2  
      RET
```

```
YS20MS: MOV R6, #06H
```


```
DL2:   MOV R7, #100
```

```
DL1:   DJNZ    R7, DL1  
      DJNZ    R6, DL2
```

```
RET
```

```
END
```

8. 系统调试运行结果说明、分析所出现得问题，设计体会与建议



9. 结论及设计体会

这次课程设计让我对单片机有了更深刻的理解和认识，在老师的讲授下学会了很多关于单片机的知识，在程序调试过程中遇到了很多困难，学习了怎么去解决问题单步调试发现错误，这样才能起到事半功倍的效果。

参考文献

《MSC-51 单片机应用实验教程》 秦晓梅主编 大连理工大学出版社